
$oninline 
/* 
 GAMS program used to estimate graph efficiency. 
 
 Author:  John B. Walden 
          NMFS/NEFSC 
          166 Water St.  
          Woods Hole, MA  02543 
          (508) 495-2355 
          John.Walden@Noaa.Gov 
 
Reference: Färe, R., S. Grosskopf and C.A.K. Lovell. 1994. Production Frontiers. 
 Cambridge University Press. 
 
 
*/ 
 
/* The following line turns off listing of some elements in the GAMS listing file*/ 
 
$OFFSYMLIST OFFSYMXREF OFFUELLIST OFFUELXREF 
 
OPTION SOLPRINT=OFF, SYSOUT=OFF, LIMROW=0, LIMCOL=0; 
 
SET INOUT /out1*out6, fix1*fix3, var1, var2/ 
    OUTPUT(INOUT) /out1*out6/ 
    INPUT(INOUT) /fix1*fix3, var1, var2/ 
 
    OBS /1*200/ 
    SUBOBS(OBS) /1*82/ 
    ACTOBS(OBS); 
 
/*We have allocated enough memory for 200 observations, but our data set only contains 82 observations (subobs)*/ 
 
 
/*Next, define an alias for the set SUBOBS */ 
 
alias (subobs, subobs1) 
 
/* The include statement below reads in an external datafile which contains a table of observations, inputs 
and outputs. The files is in comma separted value format (CSV) which can be written from, and read 
into spreadsheet programs 
 
The offlisting command means that the data won't be included in the listing file. 
 
*/ 
 
$OFFLISTING 
 
TABLE ACT(OBS,INOUT)  INPUT OUTPUT TABLE 
$ondelim 
$INCLUDE "cap1.csv" 
$offdelim 
 
$ONLISTING 
 
VARIABLES 



 
 GAMMA     efficiency score 
 weight(obs)  weights 
; 
 
 POSITIVE Variable weight; 
 
 
 
 EQUATIONS 
 
   CONSTR1(OUTPUT, OBS)  DEA constraint for each output 
   CONSTR2(INPUT, OBS)  DEA constraint for each input; 
 
   CONSTR1(OUTPUT, ACTOBS).. SUM(SUBOBS, WEIGHT(SUBOBS)*ACT(SUBOBS,OUTPUT))  
                                                               =G= ACT(ACTOBS,OUTPUT); 
 
   CONSTR2(INPUT, ACTOBS).. SUM(SUBOBS,WEIGHT(SUBOBS)*ACT(SUBOBS,INPUT))  
                                                                =L= Gamma*ACT(ACTOBS,INPUT); 
 
 
/*Define a parameter to hold results for each pass through the loop*/ 
 
 
 PARAMETER 
 
   score1(obs)  objective values 
   score2(obs)   efficiency scores 
; 
 
/*Define an external file to hold results which tell whether model solved at each iteration*/ 
 
file primal2 /te_graph_v2.txt/ 
 
/* The file te_res.txt holds information for each pass through the loop so you know the model solved at each iteration*/ 
 
 MODEL TE /ALL/; 
 
Te.solprint=2; 
te.solvelink=2; 
 
 LOOP(SUBOBS1, 
 
 
      ACTOBS(OBS)=NO; 
      ACTOBS(SUBOBS1)=YES; 
 
 
      SOLVE TE minimizing GAMMA USING LP; 
 
 
      score1(subobs1)=Gamma.l; 
      score2(SUBOBS1) = GAmma.l**(1/2); 
 
     put primal2; 
 



      if ((te.modelstat eq 1 and  te.solvestat eq 1), 
 
        put @1, subobs1.tl, @10, "optimal", @20, "normal completion" / 
 
      else 
 
        put @1, subobs1.tl, @10, te.modelstat:>2:0, 
             @20, te.solvestat:>2:0/ 
      ); 
); 
 
 
/*The next file is to output results to a file to be imported into a spreadsheet program.  Results could also be printed to 
the listing file with the use of the display command*/ 
 
 
file res /'graph_crs.csv'/ ; 
 
res.pc=5; 
res.pw=500; 
 
put res; 
 
put 'Obs','gamma', 'efficiency'; 
 
put / 
 
 loop (subobs1, 
 
    put / 
 
       put subobs1.tl,score1(subobs1), score2(subobs1); 
        ); 
putclose; 
 


